QTS QUALITY TEST SERVICES, INC.

kmowbray@qualitytest.com

Optomistic Products' Universal LightProbe Penta Trident Sensor Development for the 3070

Knut W. Mowbray

Revised 2017

Overview

The purpose of this article is to demonstrate 3070 library development techniques required to develop consistent Optomistic Products' Universal LightProbe Penta Trident Sensor color and intensity tests with any configuration LED, from a single LED to multiple LED packages. The same techniques may be used on any of the other Optomistic Products' light sensors such as the Universal LightProbe™ Spectra Sensors which measure general purpose color responses, returning an analog voltage proportional to the light's wavelength.

There are many possible methods to this solve this problem depending upon the board topology and the level of control available to light the LEDs. This approach, however, has been shown to be the most flexible and adaptable to various topologies present in board designs from dashboard electronics to network status indicator panels.

Sensor capabilities

Let us first review the capabilities of the Optomistic Products' Universal LightProbe Penta Trident Sensor color sensor. These sensors have 3 light pipe inputs, are powered by up to +40 volts, and have 2 outputs. The first output is the color output which returns a voltage depending upon which color is on any light pipe input. The second output produces a voltage proportional the the light intensity. The following table lists the colors and their voltages:

Color	Voltage
Dark	0.5 volts +/- 0.025 volts
Blue	1.0 volts +/- 0.025 volts
Green	1.5 volts +/- 0.025 volts
Yellow	2.0 volts +/- 0.025 volts
Orange	2.5 volts +/- 0.025 volts
Red	3.0 volts +/- 0.025 volts
White	3.5 volts +/- 0.025 volts

The intensity output has a range of 0.0 to 4.1 volts.



Illustration 1: Optomist Products' Universal LightProbe Penta Trident Sensor

Advantages of the Optomistic Products' Universal Light Probe Penta Sensor

The Optomistic Products probe is well adapted to ICT applications for the following reasons:

1. There is no circuit board to install in the fixture.
2. “Dummy” nodes in the board file are NOT required.
3. Node library tests are not required.
4. Digital control is not required. The tests are analog powered only.
5. Muxing and control is done entirely by the 3070 switching system.
6. The wiring that illuminates the LEDs is generated automatically by the 3070 (if you do what we recommend in this article).
7. The Trident probes may be positioned at convenient locations inside the fixture close to the LED's under test.

Designing the 3070 libraries

Our goals are to create as much re-usable library code so, in the future, we can quickly develop new Optomistic LED color check library models when new LED device topologies appear. Also we want to have meaningful sub test names, i.e. “color” measures the color output and “intensity” measures the intensity output. In addition, we need to assure that the only LED on during each LED test is the device under test. Since the Optomistic Products' Universal LightProbe Penta Trident sensors has 3 light pipe inputs, any light input on any of the 3 inputs will produce a reading. It is therefore imperative that only a single LED be on during each “color” and “intensity” test.

Consider the following circuit:



Illustration 2: Sample LED circuit

In this circuit we have a 7 channel darlington pair inverter, driving 6 LEDs. We want our LED models to measure each LED as a diode in the unpowered portion of the test program with the color and intensity measurements in the powered portion of the test program. We also want the color and intensity tests to execute without power applied to the board.

Let's begin with the simplest case, the red LED d1. Our initial setup is performed using the Part Description Editor utility. Using the utility we create 2 entries, one to measure the LED as a diode and one entry as a pin library. The resulting source code follows:

```
diode "RED", 1.9, 1.7, nr

pin library "ci", nr, pn"optomistic"

external pins 1
   device "ci" pins "Anode"
   device "RED" pins "A"

external pins 2
   device "ci" pins "Cathode"
   device "RED" pins "C"
```

We have a standard diode test mapping the cathode to pin 2 and the anode mapping to pin 1. We also reference an external pin library file named "optomistic". This is the powered portion of the test. Now let's examine the source code the the powered analog test "optomistic".

```
!
! Optomistic Products' Universal LightProbe Penta Trident sensor library
!
test powered analog

    connect s to pins "Anode"
    connect a to pins "Cathode"

    source dcv, am8, on, terminated 50, icompliance 1
    auxiliary dcv, am0, on, icompliance 1

    wait 400m  ! sensors take 300mS to 400mS to take reading

    test "color"
    test "intensity"

end test

subtest "color"
! Measurement statement will be added later
end subtest

subtest "intensity"
! Measurement statement will be added later
end subtest
```

Optomistic Products Note: Using the new "S2" Sensors, the response time will be reduced to apprx. 100mS to take a reading for each LED.

This test applies a voltage to the anode of the diode under test and uses the source's termination resistor as a current limit resistor. We use the auxiliary source as ground by setting its voltage to 0 volts which is applied to the cathode completing the circuit. We can control the current applied by varying the source's voltage and terminating resistor values. Since we cannot mix nodes names with pins names in device libraries, we cannot list the Optomistic Products' Universal LightProbe sensor outputs at this time, hence the empty subtests "color" and "intensity". We will resolve those issues later in the development process.

This library will create the 2 tests for d1 named “d1%red”, the unpowered diode test and “d1%ci”, the color and intensity tests. After the initial IPG run, the subtests “color” and “intensity” will remain empty. A simplified schematic of the 3070 setup is shown below:

Illustration 3: 3070 Simplified setup



Now let's examine a slightly more complicated LED arrangement, in this case, d2, a dual color LED (see schematic on page 2). Again using the Part Description Editor we map the first LED, we'll call “a”, anode to pin 1, cathode to pin 3 and the second LED, we'll call “b”, anode to pin 2, cathode to pin 4. Likewise we'll map 2 pin libraries “optomistic” to the matching anode and cathode pins. The completed part description library is shown below:

```
diode "a", 1.9, 1.7, nr
diode "b", 1.9, 1.7, nr

pin library "ci-a", nr, pn"optomistic"
pin library "ci-b", nr, pn"optomistic"

external pins 1
   device "a" pins "A"
   device "ci-a" pins "Anode"

external pins 2
   device "b" pins "A"
   device "ci-b" pins "Anode"

external pins 3
   device "a" pins "C"
   device "ci-a" pins "Cathode"

external pins 4
   device "b" pins "C"
   device "ci-b" pins "Cathode"
```

After our initial IPG run, this part description will create 4 tests, the unpowered diode tests “d2%a”, and “d2%b” and the powered tests “d2%ci-a” and “d2%ci-b”.

Identical techniques are used to create the part description library for the 3 color LED module d3 (See schematic on page 2). The source code follows:

```
diode "a", 1.9, 1.7, nr
diode "b", 1.9, 1.7, nr
diode "c", 1.9, 1.7, nr

pin library "ci-a", nr, pn"optomistic"
pin library "ci-b", nr, pn"optomistic"
pin library "ci-c", nr, pn"optomistic"

external pins 1
   device "a" pins "A"
   device "ci-a" pins "Anode"

external pins 2
   device "b" pins "A"
   device "ci-b" pins "Anode"

external pins 3
   device "c" pins "A"
   device "ci-c" pins "Anode"

external pins 4
   device "c" pins "C"
   device "ci-c" pins "Cathode"

external pins 5
   device "b" pins "C"
   device "ci-b" pins "Cathode"

external pins 6
   device "a" pins "C"
   device "ci-a" pins "Cathode"
```

This part description library will create 6 tests, diode tests “d3%a”, “d3%b” and “d3%c” and powered tests “d3%ci-a”, “d3%ci-b” and “d3%ci-c”.

Now we are ready for the first IPG run to create our initial executable tests. As mentioned before, the analog powered tests will have empty subtests “color” and “intensity”. After the initial IPG run we will resolve these issues. Alternatively we may select an interactive IPG run, stopping at the create requirements objects step. In either case, the following steps are necessary to complete the tests.

Resolving the empty subtests

We are measuring 6 LEDs so we will need 2 Optomistic Products' Universal LightProbe Penta Trident sensors. We must create 2 nodes for the color outputs, 2 nodes for the intensity outputs and the power and ground nodes to power the sensors. Use the fixture consultant application (Pull down: Tasks| View/Edit Fixture Electronics | Add/Delete Nodes) to create the following “terminal” nodes: “color1”, “color2”, “intensity1”, “intensity2”, “opt5V”, and “optoGND”. Re-save the fixture file. We are now ready to resolve our empty subtests “color” and “intensity”.

Our plan is to map the light pipes from our first sensor to d1, d2a and d2b using nodes “color1” and “intensity1”. The second sensor will be mapped to d3a, d3b and d3c using nodes “color2” and “intensity2”.

Each of the powered analog tests will need to be edited, re-saved and compiled. Each compilation will create a requirements object file.

The IPG output for the test “analog/d1%ci”, before editing, is shown below:

```
! IPG: rev 05.21p  Thu Mar 10 14:27:05 2011
!
! Optomistic Products' Universal LightProbe Penta Trident sensor library
!
test powered analog

    connect s to pins "Anode"
    connect a to pins "Cathode"

    source dcv, am8, on, terminated 50, icompliance 1
    auxiliary dcv, am0, on, icompliance 1

    wait 400m  ! sensors take 300mS to 400mS to take reading

    test "color"
    test "intensity"

end test

subtest "color"
! Measurement statement will be added later
end subtest

subtest "intensity"
! Measurement statement will be added later
end subtest
```

Optomistic Products Note: Using the new "S2" Sensors, the response time will be reduced to apprx. 100mS to take a reading for each LED.

We need to fill in the empty subtests “color” and “intensity”. Knowing this in advance, a temporary analog file was created which is intended to be merged in to the executable tests created by IPG. The source is listed below:

```
subtest "color"

    connect i to "colorN" ! replace N
    connect l to ground

    detector dcv, expect 5
    measure 1.25, 0.75      ! blue ...select your color...
    measure 1.75, 1.25      ! green
    measure 2.25, 1.75      ! Yellow
    measure 2.75, 2.25      ! Orange
    measure 3.25, 2.75      ! red
```

```
    measure 3.75, 3.25      ! white

end subtest

subtest "intensity"

   connect i to "intensityN"  ! replace N
   connect l to ground

   detector dcv, expect 5, rejection 1
   measure 900m, 200m

end subtest
```

Note this test fragment contains all the possible color range measurements, and 2 dummy nodes “colorN” and “intensityN”. Merge this file into each LED color test, replacing the empty subtests. Then edit the node names and delete all the measure statements except the measure statement matching the LED color. The edited version of “d1%ci” is shown below:

```
! IPG: rev 05.21p  Thu Mar 10 14:27:05 2011
!
! Optomistic Products' Universal LightProbe Penta Trident sensor library
!
test powered analog

    connect s to pins "Anode"
    connect a to pins "Cathode"

    source dcv, am8, on, terminated 50, icompliance 1
    auxiliary dcv, am0, on, icompliance 1

    wait 400m  ! sensors take 300mS to 400mS to take reading

    test "color"
    test "intensity"

end test

subtest "color"

    connect i to "color1"
    connect l to ground

    detector dcv, expect 5
    measure 3.25, 2.75      ! red

end subtest

subtest "intensity"

   connect i to "intensity1"
   connect l to ground

   detector dcv, expect 5, rejection 1
   measure 900m, 200m
```

```
end subtest
```

After the edit and re-save, compile each of the LED color tests. The final step is to run or continue IPG to resolved the requirements objects created by these edits.

Documenting the additional wiring

You will need to create a list of BRC's created in the last step to document the wiring and communicate the Optomistic Products' Universal LightProbe Penta Trident wiring requirements to your fixture vendor. The completion of IPG will assign BRC's for the measurement nodes “color1”, “intensity1”, “color2” and “intensity2”. Look up the BRC's using the fixture consultant application to acquire the information. An example is shown below:

```
Special wiring for Optomistic LED sensors
Power:
   Wire power pins (labeled "P") together, daisy chain OK.
   Transfer pin wired to BRCs:
   21309, 21310, 21312
   Wire ground pins (labeled "G") together, daisy chain OK.
   Transfer pin wired to BRC's:
   21311, 21307, 21308, 21301

   Sensors are named 01 to 02.  The light pipes must be placed
   above the listed LED's below.

   Wire the Color BRC to the indicated sensor "Color" pin.
   Wire the Int. BRC to the indicated sensor "Int" pin.

   LED   Sensor   Color BRC   Int. BRC
   ----  -----    ---------   --------
   d1    01       21403       21405
   d2a   01
   d2b   01
   d3a   02       21404       21406
   d3b   02
   d3c   02
```

Modifying the testplan

The final step is to modify the testplan file. The program generator will place all the LED color and intensity tests in the subroutine: “Analog_Functional_Tests”. These tests need to be moved to the end of the “Analog_Tests” subroutine. The edited subroutine is shown below:

```
sub Analog_Tests (Status_Code, Message$)
global Status

   if Message$ <> "" then  print tab(5);Message$
   Status = Status_Code
   test "analog/d1%red"
   test "analog/d2%a"
   test "analog/d2%b"
   test "analog/d3%a"
   test "analog/d3%b"
```

```
   test "analog/d3%c"
   test "analog/r1"
   test "analog/r2"
   test "analog/r3"
   test "analog/r4"
   test "analog/r5"
   test "analog/r6"

   powered
   cps
   dps 1              ! disconnect all board power supplies

   sps 2,5.0,0.5;optimize

   test "analog/d1%ci"
   test "analog/d2%ci-a"
   test "analog/d2%ci-b"
   test "analog/d3%ci-a"
   test "analog/d3%ci-b"
   test "analog/d3%ci-c"

   unpowered

subend
```

Mechanical considerations

Do NOT trim the the light pipes to length! If needed, custom length light pipes can be ordered from Optomistic Products. The light pipes have a minimum bend radius, so avoid small bend radii. Keep the board under test in the dark to avoid outside light source interference.

Makes sure each light pipe's distance from each LED is adjustable. When the final position is determined, fix the pipe in place with low temperature hot glue.

Troubleshooting tips

Before you begin to debug the tests, verify the wiring of each Optomistic Products' Universal LightProbe sensor color and intensity connections. Verify each sensor has the correct power (+5 volts). Verify the grounding is correct. Finally verify the light pipes are placed on the correct LEDs.

If your color value returns 0.5 volts, the LED is “dark”. First verify the LED under test is actually on visually. If the LED is not on, most likely the pin mapping of the anode and cathode was reversed. To fix this, reverse the current direction in the test source as shown below:

```
source dcv, am0, on, terminated 50, icompliance 1
auxiliary dcv, am8, on, icompliance 1
```

Another possibility is the light pipe is too far away or too close to the LED. Adjust the light pipe's distance from the LED under test until you receive the correct reading. When adjusting the light pipes, be careful not to damage the fiber, by crimping or bending the pipe. Fix the light pipe in place with low temperature hot glue when adjusted properly.

A “dark” reading may also mean the LED is too bright. A too bright LED will saturate the sensor. Lower the source voltage, or increase the series termination value to reduce the LED brightness.

If the LED is on and adjusting the light pipe makes no difference, verify the light pipe is properly seated in the sensor. You will need a special tool, available from Optomistic Products to loosen and re-seat the light pipes in the sensor.

You are now ready to have your fixture built and debug the tests. Happy coloring!

If you encounter any problems implementing these tests, please feel free to contact us for assistance.

Addendum A

This addendum is authored by Optomistic Products to address the option of utilizing the Universal LightProbe **Spectra** Sensor with Trident Fiber-optic probes with the same development techniques described in the Quality Test Services article regarding Universal LightProbe Penta Sensors.

The Universal LightProbe Penta Sensor provides “color-binning,” assigning a voltage output for each of the five main LED colors, plus white. The Universal LightProbe **Spectra** Sensor, on the other hand, will test for any LED color wavelength in the visual band (400-700nm) and also white. Both sensors provide the same intensity output with a range of 0.0 to 4.1 volts.

Universal LightProbe Spectra Sensors allow you to assign your own color boundaries when testing LEDs. Universal LightProbe Spectra Sensors check LED color by providing a voltage output proportional to LED wavelength. For colors, LED Wavelength = [100(Vout +4)nm] and for white, the voltage output is 3.5volts. This eliminates the need to convert LED wavelength as a pulse rate which effectively improves the overall response time to test an LED.

On page 1 of the QTS article, the author provides a table outlining the color and corresponding voltage output of the Penta Sensors. When using the Spectra Sensors, this Penta Sensor color table would be replaced with the following Spectra Sensor color table (as an example):

Color	**Min V_{out}**	**Max V_{out}**
Dark	0.5	0.5
Blue	0.48	0.84
Green	1.60	1.70
Yellow	1.83	1.92
Orange	2.02	2.12
Red	2.21	2.94
White	3.45	3.55

Keep in mind that these are examples, and for the Spectra Sensor, you may assign your own color boundaries (and, in fact, your own colors). However, you will not want the boundaries to "overlap" – ie: your Max V_{out} for Green should not overlap with your Min V_{out} for Yellow.

In the example given on page 7 for the subtest of color (in this case the color "red") the measure for red would be replaced with the Min/Max values that you have established for red using the Spectra Sensor. Using the example for red Min/Max values from the table above, the subtest would then look like this:

```
subtest "color"
        connect  i  to  "color1"
        connect  1  to  ground
        detector dcv,  expect 5
        measure  2.94,  2.21     ! red
```

In all other respects, the development techniques described in the article would remain the same.

You may contact either Quality Test Services or Optomistic Products for further assistance.